

A long-exposure photograph of a city at night, featuring a complex multi-level highway interchange in the foreground with prominent white and red light trails from moving vehicles. The background is filled with a dense skyline of illuminated skyscrapers, including the Shanghai Tower and the Oriental Pearl Tower, under a dark, cloudy sky.

# LUSAS

## LUSAS Programmable Interface (LPI)

Customisation and Automation Guide



# **LUSAS Programmable Interface (LPI) Customisation and Automation Guide**

---

**LUSAS Version 22.0 : Issue 1**

LUSAS  
Forge House, 66 High Street, Kingston upon Thames,  
Surrey, KT1 1HN, United Kingdom

Tel: +44 (0)20 8541 1999  
Fax +44 (0)20 8549 9399  
Email: [info@lusas.com](mailto:info@lusas.com)  
<http://www.lusas.com>

Distributors Worldwide

Copyright ©1982-2025 LUSAS  
All Rights Reserved.

## Table of Contents

<b>Introduction</b>	<b>1</b>
Introduction .....	1
Examples of capabilities .....	2
Scripts .....	2
Topics covered in this guide .....	3
LPI Developer Guide .....	3
<b>Getting started with the LUSAS Programmable Interface (LPI)</b>	<b>5</b>
LPI Command Bar .....	5
Identifying LPI functions .....	6
Writing LPI functions to a file .....	6
Writing LPI functions to the Text Output window .....	7
Details of LPI functions .....	8
LUSAS Programmable Interface (LPI) online help .....	8
Searching LPI help .....	9
Some function basics .....	10
<b>Customising the user interface</b>	<b>13</b>
Capabilities .....	13
Modifying standard toolbars .....	13
Customized toolbar buttons .....	14
User toolbar buttons .....	15
Startup templates .....	16
<b>Getting started with VBS</b>	<b>19</b>
Programming syntax .....	20
Some simple rules .....	20
Visual Basic Script online tutorials .....	21
<b>Example VB scripts</b>	<b>23</b>
Simple user script example .....	23
Deleting a range of loadcases .....	23
Running a script .....	26
Supplied script examples .....	26
Example script: Attributes.vbs .....	27
Example script: Results.vbs .....	30
Example script: Arbitrary_section.vbs .....	33
Running a script from a menu .....	35
Adding a user menu .....	36
More advanced scripts .....	36


## Table Of Contents

---

# Introduction

## Introduction

LUSAS software is highly customisable. The built-in LUSAS Programmable Interface (LPI) allows the customisation and automation of modelling and results processing tasks and creation of user-defined menu items, dialogs and toolbars as a means to access those user-defined resources. It can also be used for transferring data between LUSAS and other software applications, and to control other programs from within LUSAS Modeller, or control LUSAS Modeller from other programs.



With LPI, any user can automate the creation of complete structures, either in LUSAS or from third-party software, carrying out design checks, optimising members and outputting graphs, spreadsheets of results and custom reports. Because everything carried out by a user is recorded in a LUSAS Modeller session file, anything that LUSAS can do, can also be controlled by another application via the LUSAS Programmable Interface. This means that you can view and edit a recorded session,

parameterise those commands, turn them into sub-routines, add loops and other functions to the scripts and create a totally different application or program - using the proven core technology of LUSAS.

In addition to the accessing and customising LUSAS Modeller via the LUSAS Programmable Interface, user-defined material models (written in Fortran) can be compiled and built into a customised LUSAS Solver executable by using the LUSAS Material Model Interface (LUSAS MMI).

## Examples of capabilities

By using any ActiveX compliant scripting language, such as VB.Net, C#, VBScript, C++, Python, Perl, JScript etc. to access LUSAS facilities and functionality, you can:

- ☐ Create user-defined menu items, dialogs and toolbars
- ☐ Interrogate all aspects of a LUSAS model
- ☐ Customise modelling operations
- ☐ Create parameterised models
- ☐ Automate repetitive tasks
- ☐ Import CAD geometry and properties
- ☐ Make direct links to Microsoft Word / Excel, or other programs for import or export of data
- ☐ Perform simple / codified design checks and, when used with automated iterative analysis, optimise structural member sizes and configurations, slab reinforcement quantities, etc.

## Scripts

In their simplest form script files can be used to store a sequence of LUSAS commands for later playback. Some examples of use include the creation of start-up templates to pre-load the Attributes Treeview of the LUSAS Modeller user interface with selected attributes for a particular analysis; the setting of default mesh or material types, or preferred colour schemes; or defining specific model orientations for use when saving model views for use in reports.

When LUSAS is run, a session file is created recording each step of the model generation in Visual Basic Script (.vbs) - or with an .lvb file extension from version 20 onwards) - one of the most commonly used and easily understood languages. Editing of a session file can be used to define a similar model with new parameters. When the script is re-run in LUSAS, a new user-defined model can be easily and rapidly generated from the parameters defined. A Macro Recorder facility in LUSAS also provides the means to record a sub-set of commands for a task, for saving and re-use.



User-generated scripts can be controlled by creating dialogs that may include parametric variables, check boxes, drop-downs etc.

Varied uses of scripts include reading of geometric data, such as column dimensions, section properties and span lengths / storey heights etc., from a spreadsheet to automatically build multi-span bridge or building models; rapid generation of parametrically-idealised wind farm base structures, or for automating the creation of numerous load combinations and envelopes.

A set of example scripts are provided in LUSAS to assist in the understanding of standard concepts including file handling, how to access LUSAS geometry / attribute data, and how to import / export data from / to Microsoft Word or Excel, or other programs.

## Topics covered in this guide

The aim of this guide is to help you locate and use the supplied tools which will enable you to write scripts and work more efficiently. No programming experience or knowledge is needed to complete the examples shown. The guide covers:

- ☐ **Getting started with the LUSAS Programmable Interface (LPI)**
- ☐ **Identifying LPI Functions**
- ☐ **Customising the interface**
- ☐ **Getting started with VBS**
- ☐ **A simple example script**
- ☐ **Creating your own menus**

## LPI Developer Guide

A separate LPI Developer Guide is also available covering more advanced topics:


- ☐ **Creating dialogs using VB.NET**
- ☐ **LUSAS via COM**
- ☐ **LUSAS Material Model Interface**




# Getting started with the LUSAS Programmable Interface (LPI)

## LPI Command Bar

The LPI Command Bar can be added to the user interface by selecting the menu item **View> LPI Command Bar**




Everything that can be selected within LUSAS Modeller can be initiated by typing commands into the LPI Bar. For instance, when the **Save** button is selected in LUSAS Modeller, it is actually calling the function **database.save()**



Therefore, to save a model, you can type in the LPI command bar:

```
call database.save()
```

and then press Enter:



Commands can also be concatenated using a colon (:) character.

For example:

```
Txt = "Hello World" : call msgbox(txt)
```

## Identifying LPI functions

There are two ways of identifying which LPI function corresponds to an operation carried out within LUSAS:

- ☐ By writing LPI functions to a file
- ☐ By writing LPI functions to the Text Output window.

### Writing LPI functions to a file

1. Select the menu item **File > Script > Start Recording...**
2. Pick menu item(s), for example select **Tools > Vertical Axis** and click **OK**.
3. Select the menu item **File > Script > Stop Recording**

A .vbs file (.lvb file from version 20 onwards) will be saved to a chosen location. This file can then be edited with a text editor to see the LPI commands. An example follows:


```
$ENGINE=VBScript

' LUSAS Modeller session file
' Created by LUSAS 21.0-0c1 - Modeller Version 21.0.1601.44691
' Created at 14:17 on Thursday, September 28 2023
' (C) Finite Element Analysis Ltd 2023
'

call setCreationVersion("21.0-0c1, 21.0.1601.44691")
'

'*** Settings/Options/Properties change
call database.setVerticalDir("Y")
```

### Re-running commands with Modeller


- ☐ Commands can be re-run within Modeller by selecting **File > Script > Run Script...** and choosing the previously saved .vbs file.
- ☐ The **Run Script**  button can also be selected to run scripts.

### Writing LPI functions to the Text Output window

1. Select the menu item **File > Model Properties > Options > Advanced... > New option...**
2. Type **echocommands** and then click on **“Boolean”**. Tick the option **“Value”**.
3. Click **OK** on all dialog buttons presented.

Now, for every operation carried out within LUSAS Modeller, the corresponding command will be written to the Text Output window.

For example: If you click on the **Save** button, you will see the following:




### Details of LPI functions

Details of all LPI functions along with an explanation of what they do, the arguments they take (if any), the returned values, etc, can be found by selecting the LUSAS menu item **Help > LPI Reference Manual**

### LUSAS Programmable Interface (LPI) online help

The left-hand pane of the LUSAS Programmable Interface help system contains a filtered list of all classes and functions that can be accessed within Modeller.



Clicking one of these classes or functions gives a description of the function with its input arguments and return values in the right-hand pane. Any argument shown in square brackets is optional.

loadset
Assignment.addLoadsetSpecified
Assignment.getAssignmentLoadset
Assignment.setLoadset
Assignment.setLoadsetAll
Assignment.setLoadsetFromStart
Assignment.setLoadsetOff
Assignment.setLoadsetRange
Assignment.setLoadsetSpecified
Assignment.setLoadsetToEnd
Database.countLoadsets
Database.createLoadsetTargetValues
Database.deleteLoadset
Database.deleteLoadsets
Database.existsLoadset
Database.getLoadsetID
Database.getLoadset
Database.getLoadsets
Database.getSmallestLoadsetID
Loadset
Loadset.TargetValues
LoadsetTargetValues.addEntries
LoadsetTargetValues.addEntry
LoadsetTargetValues.addOptimisationCriteria
LoadsetTargetValues.addOptimisationCriteria
on
LoadsetTargetValues.addTarget
LoadsetTargetValues.addTarget
LoadsetTargetValues.createCombination
LoadsetTargetValues.createLoadcase
LoadsetTargetValues.factorLoadcases
LoadsetTargetValues.getConstantFactors
LoadsetTargetValues.getEigenvalueDs
LoadsetTargetValues.getFactorTypes
LoadsetTargetValues.getHarmonicDs
LoadsetTargetValues.getLoadcaseDs
LoadsetTargetValues.getOptimisationCriteria

## Loadset.createValue

createValue(name, [energy], [force], [length], [mass], [time], [temperature], [perUnitLength])

Create a new value within this loadset for subsequent use. The initial value will be 0.0 until modified by a call to [setValue](#). LUSAS will not use this value for any purpose, but will store it in model files, and allow subsequent modification with [setValue](#) and/or subsequent access with [getValue](#). The value may have any simple data type - integer, boolean, real or string, but may not be an array, object or other complex type. For numbers, it will often be desirable, but is not compulsory, to attach unit information to the value, such that its value can be fetched or modified in a known system of units. This is done using the six optional integers. The integers represent the indices, or 'power' of each scalar quantity - e.g. 2=squared, 3=cubed and so on. Each integer may be positive or negative. E.g. specifying '0.0,1,0,0,0' would mean that the new quantity is a length; '0.0,2,0,0,0' would mean length squared, i.e. area; '0.0,1,0,-1,0' would mean length divided by time, i.e. velocity; and '0.1,-2,0,0,0' would mean force per unit area.

name	string	name of the new value
energy	optional integer	energy component of the new value (default 0.0)
force	optional integer	force component of the new value (default 0.0)
length	optional integer	length component of the new value (default 0.0)
mass	optional integer	mass component of the new value (default 0.0)
time	optional integer	time component of the new value (default 0.0)
temperature	optional integer	temperature component of the new value (default 0.0)
perUnitLength	optional integer	Only to be used for quantities that are 'per unit length' or 'per unit area', such as "mm <sup>2</sup> /m" (default 0.0)
Return value	IFDispatch	

See also [setValue](#) [getValue](#) [setValueDescription](#)

[Back to Loadset](#)

[Back to Overview](#)

The creation of a new model will require a call to the [Modeller.newDatabase](#) function and from this other functions may be called to add to, manipulate or interrogate the state of the objects in the Modeller database.

modeller
Global Functions
Modeller.Dlg
ModellerDlg.getMainWndHandle
ModellerDlg.onClose
Modeller.abacusExport
Modeller.AbzMsgBox
Modeller.getAnimationManager
Modeller.anyExport
Modeller.getSessionFileAssignment
Modeller.bringToFront
Modeller.browseDir
Modeller.calculateAndWriteSection
Modeller.closeEnd
Modeller.closeStart
Modeller.closePrintFile
Modeller.getCmdFileExportData
Modeller.getCmdFileImportData

## Modeller.newDatabase

newDatabase([analysisType], [filename])

Creates a new, blank, model database without saving changes in the previous model, if any. (Use [db.save](#) first.)

analysisType	optional string	Analysis type to create: "Structural", "Thermal", or "Coupled"
filename	optional string	Filename to be created, e.g. "C:\temp\myModel.mdb". If not given, the database will not be associated with any file until <a href="#">saveAs</a> is called
Return value	IFDatabase	The new database

[Back to Modeller](#)

[Back to Overview](#)

## Searching LPI help

Searching of LPI help is possible. For example, a search for "save" will return results that include the [database.save\(\)](#) function:

save

AnalysisBaseClass.deleteSavedArray  
AnalysisBaseClass.existsSavedArray  
AnalysisBaseClass.saveStrArray  
AnimationView.saveAs  
Database.deleteSavedArray  
Database.existsSavedArray  
**Database.save**  
Database.saveAs  
Database.saveDbfArray  
Database.saveIntArray  
Database.saveStrArray  
Database.setSaveSafety  
Graph.savePicture  
GridWindow.saveAllAs  
GridWindow.saveAs  
Modeler.createSaveDialog

## Database.save

---

save()  
  
Save the model to disk.

Arguments	none
Return value	none

[Back to Database](#)  
[Back to Overview](#)

### Some function basics

Note that the above **database.save()** function it does not take any parameters and does not return anything. It simply saves the model.

The **database.saveAs(filename)** function is an example of a function that does take a parameter (in this case, just one) comprising a string with the path and file name of the new model:

save

AnalysisBaseClass.deleteSavedArray  
AnalysisBaseClass.existsSavedArray  
AnalysisBaseClass.saveStrArray  
AnimationView.saveAs  
Database.deleteSavedArray  
Database.existsSavedArray  
Database.save  
Database.saveAs  
Database.saveDbfArray  
Database.saveIntArray  
Database.saveStrArray  
Database.setSaveSafety  
Graph.savePicture  
GridWindow.saveAllAs  
GridWindow.saveAs

## Database.saveAs

---

saveAs(filename)  
  
As [save](#) but allows specification of a new name for the model

filename	string	The path and name of the model file
Return value	none	

[Back to Database](#)  
[Back to Overview](#)

To specify that you are passing a string to the function and not any other type of data, strings need to be placed between double quotes: (“ ”)

LPI command	call database.saveAs("C:\LUSAS151\Projects\MyNewModel.mdl")
-------------	---



An example of a function that takes one parameter and returns a named attribute:

```
database.createLoadingConcentrated("MyConcentratedLoad")
```

LPI command `call database.createLoadingConcentrated("MyConcentratedLoad")`

This function creates a concentrated load attribute. The parameter it takes is the name of the load attribute (for example: MyConcentratedLoad), and it returns an object of the class 'Loading Concentrated':


## Database.createLoadingConcentrated

`createLoadingConcentrated(attrName)`

Creates a concentrated structural loading attribute

<b>attrName</b>	string	name of attribute
<b>Return value</b>	IFLoadingConcentrated	newly created attribute

The corresponding entry in the Attributes Treeview is shown as follows:



At the moment this load has a value of 0 for all its components. If you want it to be a load of, say, 10 units in the X direction, you also need to use one of the functions of the Loading Concentrated class:

# Loading Concentrated

**Base Class:** [Loading](#)

**Derived Classes:** None

**Description**

Concentrated structural loading attribute

**Available Functions:**

[setConcentrated\(px, py, \[pz\], \[mx\], \[my\], \[mz\], \[loof1\], \[loof2\], \[pore\]\)](#)

LPI command	call database.createLoadingConcentrated("MyConcentratedLoad").setConcentrated(10.0,0.0)
-------------	---

As stated previously, the parameters in square brackets [ ] are optional, so do not need to be defined, hence just px and py are specified.

# Customising the user interface


## Capabilities

With LPI you can create user-defined menu items, dialogs and toolbars. Dialogs are covered in the LUSAS Programmable Interface (LPI) Developer Guide. This section covers customisation of:

- ☐ **Modifying standard toolbars**
- ☐ **Customised and User toolbar buttons**
- ☐ **Start-up templates**

### Modifying standard toolbars


Toolbars consist of buttons which can be used to drive the software.



Within Modeller these can be customised from the **View> Toolbars** menu item.


## Customising the user interface

---




### Customized toolbar buttons

Pre-programmed buttons can be added to the toolbars from the **View > Toolbars > Customize > Customize** tab, by simply dragging and dropping buttons as required.



## User toolbar buttons

User toolbar buttons can be added and programmed to carry out user defined actions by navigating to selecting **View > Toolbars > Customize > Customize tab** and then selecting **User** from the Categories list



The bitmaps on the toolbar buttons may be changed by modifying the file **C:\<LUSAS Installation Folder>\Programs\Config\userToolbar.bmp**


### Calling functions from user buttons:

The actions carried out when a button on a user tool bar is chosen are defined on the **View > Toolbars > Customize > User** dialog.

For example, if you often need to define a concentrated load of, say, 10 in X, 20 in Y, and 30 in Z, you can type the LPI function in the user button 1 text box, so that every time that button was selected, that load attribute would be created in the Attribute Treeview:


```
Call database.createLoadingConcentrated("TypicalLoad").set
Concentrated(10,20,30,0,0,0,0,0,0)
```

Or, if you often want to run a script called MyScript.vbs, then you would type: **fileopen** "C:\LUSAS Scripts\MyScript.vbs" user button 2 text box as shown below:



### Adding User buttons to toolbar menus

User buttons can be added to the toolbar menu by dragging and dropping into place.




### Startup templates

Startup templates can be used to pre-load the Attributes Treeview with selected attributes for a particular analysis, set default mesh or material types, or define preferred colour schemes - to name just a few uses.

User-defined startup templates are created by recording the setting of a variety of selections and then associating the recording with a template name.

You can use any VBS file as a template, and you can also add templates from the "New Model" form:



Now every time that you create a new blank model, you will be able to choose this template, which will run the script just after creating the model.

## Advanced operations

For more advanced operations a macro facility is available to enable commonly used commands to be grouped together or abbreviated.

Macro functions should be written in Visual Basic and saved in a file. For example:

```
sub dp (x, y, z)

call database.createPoint (x, y, z)

end sub
```

The macro file is registered from the **Advanced** button on the LPI command bar.



The functions in the macro file may then be activated from the LPI command bar by typing the function name and arguments e.g. **dp 1,2,3** or **call dp(1,2,3)**





# Getting started with VBS


LUSAS Modeller records every operation that it carries out in a session file. This file contains standard calls to LUSAS LPI function in a Visual Basic Script syntax. The file can be replayed to carry out exactly the same actions again. Alternatively the file can be modified to carry out different actions.

The procedure is as follows:

1. To start by recording a script select **File > Script > Start Recording...**
2. Carry out a series of operations.
3. Stop recording by selecting **File > Script > Stop Recording**
4. Edit the .vbs file (.lvb file from version 20 onwards) to cover the cases required.

With a little programming syntax knowledge **loops** can be used to make the script more “tidy” and **variables** can be added to make the script more “flexible”

Editing can be carried out with the standard Windows Notepad (accessible from **Start> All Programs> Accessories> Notepad**) or 3<sup>rd</sup> party products such as Notepad++



```

MyScript.vbs - Notepad
File Edit Format View Help
$ENGINE=VBScript
' LUSAS Modeller session file
' Created by LUSAS 15.1-2c1 - Modeller Version 15.1.1617.22304
' Created at 14:52 on Tuesday, February 02 2016
' (C) Finite Element Analysis Ltd 2016
call setCreationVersion("15.1-2c1, 15.1.1617.22304")

'*** LPI Command Bar
call database.createLoadingConcentrated("TypicalLoad").setConcentrated(10.0,2
maxvalue = 20 'maxvalue is a variable that is being assigned the value of 20
For i = 1 to maxvalue
    MsgBox "Reached " & i & " out of " & maxvalue
Next
  
```

### Programming syntax

#### Some simple rules

- First line of the visual basic script file must be **\$ENGINE=VBSCRIPT**
- Lines to be treated as comments only must start with an apostrophe (')

#### Basic Operators allowed include:

- Arithmetic: +, -, /, \*
- Comparison: =, >, <, >=, <=, <>
- Concatenation: &
- Logical: Not, And, Or

#### Conditionals

- If ... Then ... Else

```
If a > b Then
    MsgBox "a was greater than b"
Else
    MsgBox "a was not greater than b"
End If
```

- Loops: For ... Next

```
maxvalue = 20 ' max value is a variable that is assigned the value of 20
For i = 1 to maxvalue
    MsgBox "Reached " & i & "out of " & maxvalue
Next
```

#### Variables

- Can be strings, numbers etc
- Names must begin with a letter
- Names must not contain an embedded full-stop (period) “.”
- Names must not exceed 255 characters
- Names must be unique
- There is no need to “declare” variables

## Arrays

- Can contain strings or numbers etc
- Can be “called” individually
- Always use (0) as the first index of the array.

Example:

```
Dim MyArray(2)
MyArray(0) = 10
MyArray(1) = 20
MyArray(2) = 30
```

This is a one dimensional array with 3 items. The first element has been assigned a value of 10, the second 20, and the third 30.

Note that the **UBound** function returns the largest available subscript of an array:

```
call msgbox("Upper bound of array=" & UBound(MyArray))
```

So the UBound function returns 2 for the array in the example above.

## Visual Basic Script online tutorials

More detailed online tutorials showing how to write visual basic script can be found on the internet. Here are just a few examples (with links applicable at the time of writing):

- English: <http://www.tizag.com/vbscriptTutorial/>
- English: <http://www.tutorialspoint.com/vbscript/index.htm>
- Chinese: <http://www.w3school.com.cn/vbscript/>




# Example VB scripts

## Simple user script example

### Deleting a range of loadcases

A simple user script is to be written to illustrate how a range of loadcases can be deleted from the Analyses Treeview.

In Version 15.2 of LUSAS the only way to delete loadcases via the user interface was to click on each of them in turn in the Analyses Treeview and press the delete key. In later versions of LUSAS this is no longer the case, and loadcases can be deleted from the Analyses Treeview singly or by multiple selection, so the script example does not provide any more functionality for these versions. However, it does provide a good introduction to generating user scripts.



### First make a recording

1. Choose **File > Start Recording...**
2. Specify a file name
3. Delete **loadcase 1** manually.
4. Choose **File > Stop Recording**
5. Open the script file created:

## Example VB scripts

---

```
1 $ENGINE=VBScript
2 ' LUSAS Modeller session file
3 ' Created by LUSAS 15.1-2c1 - Modeller Version 15.1.1617.22304
4 ' Created at 09:21 on Wednesday, February 03 2016
5 ' (C) Finite Element Analysis Ltd 2016
6 '
7 call setCreationVersion("15.1-2c1, 15.1.1617.22304")
8 '
9
10 '*** Delete loadcase/control
11
12 call database.deleteLoadset("Loadcase 1")
```

In this file note that:

- Line 1: This line is common in all scripts. Do not remove or modify this line.
- Line 7: Specifies the version of LUSAS used to generate the script. This line is common in all scripts. Do not remove or modify this line.
- Line 12: This is the line that deletes Loadcase 1
- Lines other than those above: These lines are comments: they are ignored. Comments always start with an apostrophe (‘)

If the Loadcase names are of the form Loadcase 1, Loadcase 2 etc and you wanted to delete Loadcase 2 to Loadcase 50 you need to edit the previous script and insert a **For ... Next** loop as seen at the bottom of this next image:

```
1 $ENGINE=VBScript
2 ' LUSAS Modeller session file
3 ' Created by LUSAS 15.1-2c1 - Modeller Version 15.1.1617.22304
4 ' Created at 09:21 on Wednesday, February 03 2016
5 ' (C) Finite Element Analysis Ltd 2016
6 '
7 call setCreationVersion("15.1-2c1, 15.1.1617.22304")
8 '
9
10 '*** Delete loadcase/control
11
12 For i = 2 To 50
13     call database.deleteLoadset("Loadcase " & i)
14 Next
```

Now you are actually calling the deleteLoadset function 49 times, taking the argument Loadcase 2, Loadcase 3, Loadcase 4, etc.



**Note.** The function **deleteLoadset** is used instead of **deleteLoadcase** because the deleteLoadset function also deletes combinations and envelopes.

If you look for information about this function in the LPI online help, you will see that the loadset can be specified by Loadset name (which is how it has been done in this example) but it can also be specified by Loadset ID.

## Database.deleteLoadset

---

```
deleteLoadset(loadset)
```

```
deleteLoadset(name, [resFile], [eigen], [harm])
```

```
deleteLoadset(ID, [resFile], [eigen], [harm])
```

Delete the specified loadset. Note that it is not possible to delete results loadcases (close the file instead) or the last remaining pre-processing loadcase. The loadset can be specified in several ways, by name, by ID or by type and name/ID. In each case, additionally specifying the results file name/ID, eigenvalue ID and harmonic ID will clarify to LUSAS which loadset is required. Alternatively, an object may be passed in, which requires no further clarification. This same principle applies to all functions that input single loadsets, and the examples below reflect this. Each input form is legal in each circumstance

By ID it would be easier to write the script as follows:

```
12 For i = 2 To 50
13     call database.deleteLoadset(i)
14 Next
```

And if in the original model you wanted to delete all the even loadcases you would add 'Step 2' to line 12:

```
12 For i = 2 to 100 Step 2
13     call database.deleteLoadset(i)
14 Next
```

Alternatively if you wanted to delete all the even-numbered loadcases you could append '2T50I2') to the main LPI command:

```
call database.deleteLoadsets(2T50I2)
```

This would delete from Loadcase 2 to Loadcase 50 in increments of 2.

Save the file as delete\_loadcases.vbs

### Running a script



















A script can be run within LUSAS Modeller as follows:

1. Choose **File > Script > Run Script**
2. Browse for and select **<script\_name.vbs>**

### Supplied script examples

LUSAS supplies many script examples (that are installed as part of a software installation) which demonstrate how to carry out various functions and tasks. These may be found at this location:

**C:\<LUSAS Installation Folder>\Programs (x86)\Scripts\LPIExamples**

Search LPIExamples				
Name	Date modified	Type	Size	
 Attributes.vbs	16/09/2023 00:57	VBScript Script File	4 KB	
 Dialogs.vbs	16/09/2023 00:57	VBScript Script File	2 KB	
 Display.vbs	16/09/2023 00:57	VBScript Script File	3 KB	
 ExportExcel.vbs	16/09/2023 00:57	VBScript Script File	3 KB	
 ExportWord.vbs	16/09/2023 00:57	VBScript Script File	3 KB	
 FileHandling.vbs	16/09/2023 00:57	VBScript Script File	1 KB	
 Geometry.vbs	16/09/2023 00:57	VBScript Script File	2 KB	
 Graphs.vbs	16/09/2023 00:57	VBScript Script File	2 KB	
 Groups.vbs	16/09/2023 00:57	VBScript Script File	1 KB	
 ImportExcel.vbs	16/09/2023 00:57	VBScript Script File	2 KB	
 LoadsetsResults.vbs	16/09/2023 00:57	VBScript Script File	3 KB	
 LPIExampleMenu.vbs	16/09/2023 00:57	VBScript Script File	2 KB	
 Menus.vbs	16/09/2023 00:57	VBScript Script File	2 KB	
 Mesh.vbs	16/09/2023 00:57	VBScript Script File	2 KB	
 Results.vbs	16/09/2023 00:57	VBScript Script File	3 KB	
 Selection.vbs	16/09/2023 00:57	VBScript Script File	2 KB	
 TextMessages.vbs	16/09/2023 00:57	VBScript Script File	1 KB	
 UserResults.vbs	16/09/2023 00:57	VBScript Script File	2 KB	



**Note.** In the printed versions of the two scripts that follow word wrapping has taken place. Only lines that are preceded by an apostrophe (') are comment lines. Other lines containing VB script should not be word-wrapped.



**Example script: Attributes.vbs**

This supplied script creates a single planar surface and then creates and assigns to that surface: a regular mesh, material and geometry. A support is created and assigned to a line; a concentrated load is created and assigned to a point as loadcase 1; and a face load is created and assigned to a line as loadcase 2.

```
$ENGINE=VBScript

' Create and assign attributes
'-----
-

' Create new database
call newdatabase()

' Create Surface
call geometryData.setAllDefaults()
call geometryData.setCreateMethod("planar")
call geometryData.addCoords(0.0, 0.0, 0.0)
call geometryData.addCoords(40.0, 0.0, 0.0)
call geometryData.addCoords(40.0, 20.0, 0.0)
call geometryData.addCoords(0.0, 20.0, 0.0)
call database.createSurface(geometryData)

' Create Attribute : Surface Mesh 1
call database.createMeshSurface("Plane
Stress").setRegular("QPM8", 0, 0, false)

' Modify selection
call selection.add("Surface", "1")

' Attribute : Plane Stress : Assign to Primary selection :
call assignment.setAllDefaults().setLoadset("Loadcase 1")
call database.getAttribute("Mesh", "Plane
Stress").assignTo(selection, assignment)
call database.updateMesh()

' Create Attribute : Isotropic Material 1
```

## Example VB scripts

---

```
set attr = database.createIsotropicMaterial("Mild Steel",
200.0E3, 0.3, 7.8E3)
set attr = nothing

' Attribute : Mild Steel : Assign to Primary selection :
call assignment.setAllDefaults()
call database.getAttribute("Material", "Mild
Steel").assignTo(selection, assignment)

' Create Attribute : Surface Geometric 1
call
database.createGeometricSurface("Thickness=1").setSurface(1.0,
0.0)

' Attribute : Thickness=1 : Assign to Primary selection :
call assignment.setAllDefaults()
call database.getAttribute("Geometric",
"Thickness=1").assignTo(selection, assignment)

' Create Attribute : Fixed in XY
call database.createSupportStructural("Fixed in
XY").setStructural("R", "R")

' Modify selection
call selection.add("Line", "4")

' Attribute : Fixed in XY : Assign to Primary selection :
call
assignment.setAllDefaults().setSelectionNone().addToSelection(
"Line")
call database.getAttribute("Supports", "Fixed in
XY").assignTo(selection, assignment)

' Create Attribute : Concentrated Load 1
call database.createLoadingConcentrated("Concentrated Load
1").setConcentrated(0.0, -100.0)
```

```
' Modify selection
call selection.add("Point", "3")

' Attribute : Concentrated Load 1 : Assign to Primary
selection :
call
assignment.setAllDefaults().setSelectionNone().addToSelection(
"Point").setLoadset("Loadcase 1")
' assign load to selected point in loadcase 1
call database.getAttribute("Loading", "Concentrated Load
1").assignTo(selection, assignment)

' Define face load
call database.createLoadingFace("Distributed 1").setFace(0.0,
10.0, 0.0)
' Select top face
call selection.remove("All")
call selection.add("Line", "3")

' Create new loadcase and set active
call database.createLoadcase("Loadcase 2", "Structural")
set loadset = database.getLoadset("Loadcase 2", "model")
call view.setActiveLoadset(loadset)
set loadset = nothing

' set assignment object with selected face hof and loadcase 2
set hof0 = database.getObject("Surface", "1")
call assignment.setAllDefaults().setLoadset("Loadcase
2").addHof(hof0)
' assign face load to top face in loadcase 2
call database.getAttribute("Loading", "Distributed
1").assignTo(selection, assignment)
```

### Example script: Results.vbs

This supplied script writes a set of results to the text window for a prior selection of features made in the Modeller view window.

```
$ENGINE=VBScript

' Extracting Results
' -----

    set textWindow = getTextWindow()

' set results type
    entity="Displacement"

' get array of results component names
    component=view.getResultsComponentNames(entity)

' extract array of selected nodes
    nodes=selection.getObjects("Node","All")

' check nodes in selection
    if ubound(nodes) >= 0 then

' loop selected nodes
        for i = 0 to ubound(nodes)

' get node object
            set node = nodes(i)

' get element Number
            num=node.getID()

' write line to text window
            text = "Node=" & num

            for j=0 to ubound(component)

' get averaged nodal result
                res=node.getResults(entity,component(j))

                text=text & " " & component(j) & " = " & res
```

```

        next
        textWindow.WriteLine(text)
    next
' set results type
    entity ="Force/Moment - Thick Shell"
' get array of results component names
    component=view.GetResultsComponentNames(entity)
' extract array of selected elements
    elements= selection.GetObjects("Element","All")
' loop selected elements
    for i = 0 to ubound(elements)
' set element object
        set elt = elements(i)
' get element Number
        num=elt.GetID()
' extract array of element nodes
        nodes = elt.GetNodes()
' loop element nodes
        for k = 0 to ubound(nodes)
' get node object
            set node = nodes(k)
' extract node data
            nnum = node.GetID()
            node.GetXYZ x,y,z
' get vector of element nodal results
            vecRes=elt.GetNodeVectorResults(k,entity)
            nc=ubound(vecRes)

```

## Example VB scripts

---

```
' build output text string

    text = "Elt=" & num & " Node=" & nnum & " x=" & x & " y="
& y & " z=" & z

    for j = 0 to nc
        text=text & " " & component(j) & " = " & vecRes(j)
    next

    textWindow.WriteLine(text)

next

' extract number of Gauss points

    ngp = elt.countGaussPoints()

' loop gauss points

    for k = 0 to ngp-1

        vecRes = elt.getGaussVectorResults(k,entity)

        nc=ubound(vecRes)

' build output text string

        text = "Elt=" & num & " GP=" & k

        for j = 0 to nc
            text=text & " " & component(j) & " = " & vecRes(j)
        next

        textWindow.WriteLine(text)

    next

next

else

    AfxMsgBox "Run an analysis and select some nodes as input to
this script"

end if
```

**Example script: Arbitrary\_section.vbs**

This supplied script allows creation of an Arbitrary Section

```
$ENGINE=VBScript

'*** We need a separate variable to represent the main model.
'*** This is because "db" (whilst appearing to be a variable) is
truly a
'*** function that returns the "current" database, and the
"current"
'*** database is about to be the temporary ASPC database
set maindb = db

'*** Create a special temporary database to contain all the
geometry and attributes of the new section
set aspc = newASPCDatabase()

'*** "db" and "aspc" now both refer to the temporary ASPC
database
'*** It is good practice to avoid using "db" as it may confuse
other programmers

'*** Create a surface
call geometryData.setAllDefaults()
call geometryData.setCreateMethod("planar")
call geometryData.addCoords(6.0, 19.0, 0.0)
call geometryData.addCoords(22.0, 19.0, 0.0)
call geometryData.addCoords(22.0, 28.0, 0.0)
call geometryData.addCoords(6.0, 28.0, 0.0)
call geometryData.setLowerOrderGeometryType("coordinates")
```

## Example VB scripts

---

```
call aspc.createSurface(geometryData)

'*** Optionally at this point we could assign materials, mesh,
line thickness, etc

'*** following the usual conventions of the LPI

'*** Provide the default mesh
call aspc.updateMesh()

'*** Convert the whole of the ASPC database into an ASPC utility
in the main model

call
maindb.createArbitrarySection("arbitrary_section").fromDatabase(a
spc)

'*** Save the aspc database in case we need to modify it in
future
call aspc.saveas("C:\temp\aspc.mdl")

'*** We no longer want the aspc database
call aspc.close()

'*** "db" and "maindb" now both refer to the main database

It is important to note that sections that have been created this
way cannot directly be modified in the LUSAS UI.

However they can be modified from the LPI, in exactly the same
way, just starting with the saved section:

'*** Retrieve the ASPC model we saved earlier
```



```
set aspc = openASPCDatabase("C:\temp\aspc.mdl")

'*** make some changes here

'*** Convert the whole of the ASPC database into an ASPC utility
in the main model

call
maindb.createArbitrarySection("arbitrary_section").fromDatabase(a
spc)

'*** Save the aspc database in case we need to modify it in
future

call aspc.saveas("C:\temp\aspc.mdl")

'*** We no longer want the aspc database

call aspc.close()
```

## **Running a script from a menu**


When a number of related scripts have been created it is often more convenient to add a Modeller menu item to access those scripts, rather than by opening the scripts using the **File > Script > Run Script** menu item.

To allow this, one supplied script contains the code to create a menu containing all of the other supplied scripts. To add the menu item to Modeller's main menu:

1. Choose **File > Script > Run Script**
2. Browse to the **C:\<LUSAS Installation Folder>\Programs (x86)\scripts\LPIExamples folder**
3. Select **LPIExamplesMenu.vbs**

## Example VB scripts

---



This menu script example adds a Test menu name to the main menu, and has a number of menu items with sub-menus that each trigger a script.

### Keeping the LPI menu visible

To keep the LPI menu visible:

1. Open the file **C:\<LUSAS Installation Folder>\Programs (x86)\Config\afterNewModel.vbs**
2. Add these lines to the bottom of the file:

```
Scripts= getSystemString("scripts")  
call fileopen(scripts&"\LPIExamples\LPIExampleMenu.vbs")
```

### Adding a user menu

A user menu can be added by editing this file in your user folder:

```
"%USERPROFILE%\Documents\LUSAS210\UserScripts\" -> C:\User\<Your  
username>\Documents\LUSAS210\UserScripts\Usermenu.vbs
```

## More advanced scripts

Many more advanced scripts can be downloaded from the protected User Area of the LUSAS website at:

<http://www.lusas.com/protected/download/scripts.html>





# LUSAS

LUSAS, Forge House, 66 High Street, Kingston upon Thames, Surrey, KT1 1HN, UK  
Tel: +44 (0)20 8541 1999 | Fax: +44 (0)20 8549 9399 | [info@lusas.com](mailto:info@lusas.com) | [www.lusas.com](http://www.lusas.com)